# $Id$
# Maintainer: Maxime Gauduin <alucryd@archlinux.org>

pkgname=libretro-shaders-cg
pkgver=826
pkgrel=1
pkgdesc='Collection of Cg shaders for libretro'
arch=('any')
url='http://www.libretro.com/'
license=('GPL3')
provides=('libretro-shaders')
conflicts=('libretro-shaders')
depends=('nvidia-cg-toolkit')
makedepends=('git')
_commit='a8ab0bf60fa556805cc38082b26e455406b8d0ce'
source=("libretro-shaders-cg::git+https://github.com/libretro/common-shaders.git#commit=${_commit}")
sha256sums=('SKIP')

pkgver() {
  cd libretro-shaders-cg

  git rev-list --count HEAD
}

package() {
  install -dm 755 "${pkgdir}"/usr/share/libretro
  cp -dr --no-preserve='ownership' libretro-shaders-cg "${pkgdir}"/usr/share/libretro/shaders
  rm -rf "${pkgdir}"/usr/share/libretro/shaders/.git
}

# vim: ts=2 sw=2 et:
